Problem Set 7

Math 118: Dynamical Systems

Matthew Leingang, CA

■ Initialization Cells

```
In[1]:= << Graphics`PlotField`
```

■ 1. For each of the following functions $v : \mathbb{R}^2 \to \mathbb{R}^2$, find the fixed points of $\frac{d}{dt}\begin{pmatrix} x \\ y \end{pmatrix} = v\begin{pmatrix} x \\ y \end{pmatrix}$, determine their stability, sketch.

■ (a) $v\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} x^3 - y \\ x + y \end{pmatrix}$

```
In[2]:= v[{x_, y_}] :=
         {x^3 - y, x + y}
```

First, we draw the vector field.

```
In[3]:= PlotVectorField[v[{x, y}], {x, -1, 1}, {y, -1, 1}]
```



```
Out[3]= - Graphics -
```

Based on this diagram, it looks like we have a single repelling focus (a *focus* is a fixed point where nearby points spiral out).

```
In[4]:= Solve[v[{x, y}] == {0, 0}, {x, y}]

Out[4]= {{y → 0, x → 0}, {y → -I, x → I}, {y → I, x → -I}}

In[5]:= fixedpoint = %[[1]]

Out[5]= {y → 0, x → 0}

In[6]:= Dv = ( D[v[{x, y}][[1]], x]  D[v[{x, y}][[1]], y] )
             ( D[v[{x, y}][[2]], x]  D[v[{x, y}][[2]], y] )
```

$$Out[6]= \{\{3x^2, -1\}, \{1, 1\}\}$$

```
In[7]:= Dvp = % /. fixedpoint

Out[7]= {{0, -1}, {1, 1}}

In[8]:= Eigenvalues[%]
```

$$Out[8]= \{(-1)^{1/3}, -(-1)^{2/3}\}$$

```
In[9]:= N[%]

Out[9]= {0.5 + 0.866025 I, 0.5 - 0.866025 I}
```

Since these eigenvalues are complex, we see spiraling. Since the real parts are positive, we spiral *away*.

■ (b) $v\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} \sin y \\ x + y \end{pmatrix}$

```
In[10]:= v[{x_, y_}] :=
          {Sin[y], x + y}
```

First, we draw the vector field.

```
In[11]:= PlotVectorField[v[{x, y}], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}]
```



```
Out[11]= - Graphics -
```

Let’s get a closer look.

```
In[12]:= PlotVectorField[v[{x, y}], {x, -1, 1}, {y, -1, 1}]
```



```
Out[12]= - Graphics -
```

```
In[13]:= PlotVectorField[v[{x, y}], {x, π - 1, π + 1}, {y, -π - 1, -π + 1}]
```



```
Out[13]= - Graphics -
```

To solve $v\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \end{pmatrix}$, we see that y must be a multiple of π, and x must be the opposite of y.

```
In[14]:= Dv = ( D[v[{x, y}][[1]], x]  D[v[{x, y}][[1]], y]
                D[v[{x, y}][[2]], x]  D[v[{x, y}][[2]], y] )

Out[14]= {{0, Cos[y]}, {1, 1}}

In[15]:= % /. {y -> 0, x -> 0}

Out[15]= {{0, 1}, {1, 1}}

In[16]:= Eigenvalues[%]
```

Out[16]= $\left\{\frac{1}{2}\left(1-\sqrt{5}\right), \frac{1}{2}\left(1+\sqrt{5}\right)\right\}$

```
In[17]:= N[%]

Out[17]= {-0.618034, 1.61803}
```

So at the fixed point (0,0), we see a saddle.

```
In[18]:= Dv /. {y -> π, x -> -π}

Out[18]= {{0, -1}, {1, 1}}

In[19]:= Eigenvalues[%]
```

Out[19]= $\{(-1)^{1/3}, -(-1)^{2/3}\}$

```
In[20]:= N[%]

Out[20]= {0.5 + 0.866025 I, 0.5 - 0.866025 I}
```

And at (π,$-\pi$), we see a repelling focus. Since sin is periodic modulo 2π, we have described all the fixed points.

■ (c) $v\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 3y - e^x \\ 2x - y \end{pmatrix}$

```
In[21]:= v[{x_, y_}] :=
          {3 y - ⅇ^x, 2 x - y}
```

First, we draw the vector field.

```
In[22]:= PlotVectorField[v[{x, y}], {x, 0, 3}, {y, 0, 3}]
```



```
Out[22]= - Graphics -
```

Based on this diagram, it looks like we have a saddle node.

```
In[23]:= Solve[2 x - y == 0, y]

Out[23]= {{y → 2 x}}

In[24]:= Solve[6 x - e^x == 0 , x]

        InverseFunction::ifun :
         Warning: Inverse functions are being used. Values may be lost for multivalued inverses.

        Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found.
```

Out[24]= $\left\{\left\{x \to -\text{ProductLog}\left[-\frac{1}{6}\right]\right\}, \left\{x \to -\text{ProductLog}\left[-1, -\frac{1}{6}\right]\right\}\right\}$

Okay, so that doesn't work. We have to be clever. Let's graph the function.

```
In[25]:= Plot[6 x - Exp[x], {x, 0, 3}]
```




```
Out[25]= - Graphics -
```

There are the only roots of the function, as a simple Rolle's theorem test will tell you. We can use **FindRoot** to get these roots more precisely. **FindRoot** just does Newton's method!

```
In[26]:= sol1 = FindRoot[6 x == Exp[x], {x, 0}]

Out[26]= {x → 0.204481}

In[27]:= sol2 = FindRoot[6 x == Exp[x], {x, 3}]

Out[27]= {x → 2.83315}
```

In[28]:= Dv = $\begin{pmatrix} \text{D[v[\{x, y\}][[1]], x]} & \text{D[v[\{x, y\}][[1]], y]} \\ \text{D[v[\{x, y\}][[2]], x]} & \text{D[v[\{x, y\}][[2]], y]} \end{pmatrix}$

Out[28]= $\{\{-E^x, 3\}, \{2, -1\}\}$

```
In[29]:= Eigenvalues[Dv]
```

Out[29]= $\left\{\frac{1}{2}\left(-1 - E^x - \sqrt{25 - 2 E^x + E^{2x}}\right), \frac{1}{2}\left(-1 - E^x + \sqrt{25 - 2 E^x + E^{2x}}\right)\right\}$

In[30]:= Plot$\left[\left\{\frac{1}{2}\left(-1 - E^x - \sqrt{25 - 2 E^x + E^{2x}}\right), \frac{1}{2}\left(-1 - E^x + \sqrt{25 - 2 E^x + E^{2x}}\right)\right\}, \{x, 0, 3\}\right]$




```
Out[30]= - Graphics -
```

It looks like for the smaller root of $e^x = 6x$, we have one positive and one negative eigenvalue, and is hence a hyperbolic repelling point. The larger one has two negative eigenvalues, and is hence an attractor. We can also do this numerically.

```
In[31]:= Dv /. sol1

Out[31]= {{-1.22689, 3}, {2, -1}}

In[32]:= Eigenvalues[%]

Out[32]= {-3.56556, 1.33867}

In[33]:= Dv /. sol2

Out[33]= {{-16.9989, 3}, {2, -1}}

In[34]:= Eigenvalues[%]

Out[34]= {-17.3655, -0.633375}
```

Now that we know the locations of these roots, we can zoom in on them in our plots.

```
In[35]:= x0 = x /. sol1

Out[35]= 0.204481

In[36]:= PlotVectorField[v[{x, y}],
          {x, x0 - 1 / 2, x0 + 1 / 2},
          {y, 2 x0 - 1 / 2, 2 x0 + 1 / 2}]
```



```
Out[36]= - Graphics -

In[37]:= x0 = x /. sol2

Out[37]= 2.83315
```

```
In[38]:= PlotVectorField[v[{x, y}],
          {x, x0 - 1/4, x0 + 1/4},
          {y, 2 x0 - 1/4, 2 x0 + 1/4}]
```



```
Out[38]= - Graphics -
```

■ (d) $v\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 2x + y \\ x + y \end{pmatrix}$

```
In[39]:= v[{x_, y_}] :=
          {2 x + y, x + y}
```

First, we draw the vector field.

```
In[40]:= PlotVectorField[v[{x, y}], {x, -1, 1}, {y, -1, 1}]
```



```
Out[40]= - Graphics -
```

Based on this diagram, it looks like the fixed point is repelling in all directions.

```
In[41]:= Solve[v[{x, y}] == {0, 0}, {x, y}]

Out[41]= {{x → 0, y → 0}}

In[42]:= fixedpoint = %[[1]]

Out[42]= {x → 0, y → 0}
```

In[43]:= Dv = $\begin{pmatrix} \text{D[v[\{x, y\}][[1]], x]} & \text{D[v[\{x, y\}][[1]], y]} \\ \text{D[v[\{x, y\}][[2]], x]} & \text{D[v[\{x, y\}][[2]], y]} \end{pmatrix}$

```
Out[43]= {{2, 1}, {1, 1}}

In[44]:= Dvp = % /. fixedpoint

Out[44]= {{2, 1}, {1, 1}}

In[45]:= Eigenvalues[%]
```

Out[45]= $\left\{\frac{1}{2}\left(3-\sqrt{5}\right), \frac{1}{2}\left(3+\sqrt{5}\right)\right\}$

```
In[46]:= N[%]

Out[46]= {0.381966, 2.61803}
```

Both eigenvalues are real and positive, so we see a repelling in all directions.

■ 2. Use a Lyapunov function to show that $\begin{pmatrix} -1 \\ 0 \end{pmatrix}$ is a stable fixed point for $\frac{d}{dt}\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} y^2 - x - 1 \\ -x\,y - y \end{pmatrix}$.

First, let's plot the vector field.

```
In[47]:= v[{x_, y_}] := {y^2 - x - 1, -x y - y}
```

```
In[48]:= PlotVectorField[v[{x, y}], {x, -2, 0}, {y, -1, 1}]
```



```
Out[48]= - Graphics -
```

```
In[49]:= Dv = ( D[v[{x, y}][[1]], x]  D[v[{x, y}][[1]], y] )
              ( D[v[{x, y}][[2]], x]  D[v[{x, y}][[2]], y] )
```

```
Out[49]= {{-1, 2 y}, {-y, -1 - x}}
```

```
In[50]:= % /. {x -> -1, y -> 0}
```

```
Out[50]= {{-1, 0}, {0, 0}}
```

Ah, this is trouble. We don't know what to do if we have an eigenvalue equal to zero. Linearizing the system is not an option.

But if we notice that the expressions in *v* are no worse than quadratic, we can guess at a Lyapunov function.

```
In[51]:= L[{x_, y_}] := (x + 1)^2 + y^2
```

```
In[52]:= L[{-1, 0}]

Out[52]= 0

In[53]:= lya = Plot3D[L[{x, y}], {x, -2, 0}, {y, -1, 1},
          Mesh -> False]
```




```
Out[53]= - SurfaceGraphics -
```

Let's take the time derivative of *L* along an integral curve of *v*.

```
In[54]:= Dt[L[{x, y}], t]

Out[54]= 2 (1 + x) Dt[x, t] + 2 y Dt[y, t]

In[55]:= % /. {Dt[x, t] -> y^2 - x - 1, Dt[y, t] -> -x y - y}
```

$$\text{Out[55]= } 2\,y\,(-y - x\,y) + 2\,(1 + x)\,(-1 - x + y^2)$$

```
In[56]:= Simplify[%]
```

$$\text{Out[56]= } -2\,(1 + x)^2$$

Aha! This is always negative. So we do have a Lyapunov function.

```
In[57]:= icpts = Table[
           {-1, 0} + {Cos[θ], Sin[θ]},
           {θ, 0, 2 π, π / 8}]
```

$$\text{Out[57]= } \left\{\{0, 0\}, \left\{-1 + \text{Cos}\left[\frac{\pi}{8}\right], \text{Sin}\left[\frac{\pi}{8}\right]\right\}, \left\{-1 + \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-1 + \text{Cos}\left[\frac{3\pi}{8}\right], \text{Sin}\left[\frac{3\pi}{8}\right]\right\}, \{-1, 1\},\right.$$
$$\left\{-1 + \text{Cos}\left[\frac{5\pi}{8}\right], \text{Sin}\left[\frac{5\pi}{8}\right]\right\}, \left\{-1 - \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \left\{-1 + \text{Cos}\left[\frac{7\pi}{8}\right], \text{Sin}\left[\frac{7\pi}{8}\right]\right\}, \{-2, 0\},$$
$$\left\{-1 + \text{Cos}\left[\frac{9\pi}{8}\right], \text{Sin}\left[\frac{9\pi}{8}\right]\right\}, \left\{-1 - \frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \left\{-1 + \text{Cos}\left[\frac{11\pi}{8}\right], \text{Sin}\left[\frac{11\pi}{8}\right]\right\}, \{-1, -1\},$$
$$\left.\left\{-1 + \text{Cos}\left[\frac{13\pi}{8}\right], \text{Sin}\left[\frac{13\pi}{8}\right]\right\}, \left\{-1 + \frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \left\{-1 + \text{Cos}\left[\frac{15\pi}{8}\right], \text{Sin}\left[\frac{15\pi}{8}\right]\right\}, \{0, 0\}\right\}$$

```
In[58]:= N[icpts]

Out[58]= {{0, 0}, {-0.0761205, 0.382683}, {-0.292893, 0.707107}, {-0.617317, 0.92388},
          {-1., 1.}, {-1.38268, 0.92388}, {-1.70711, 0.707107}, {-1.92388, 0.382683}, {-2., 0},
          {-1.92388, -0.382683}, {-1.70711, -0.707107}, {-1.38268, -0.92388}, {-1., -1.},
          {-0.617317, -0.92388}, {-0.292893, -0.707107}, {-0.0761205, -0.382683}, {0, 0}}

In[59]:= solns =
          Map[NDSolve[{x'[t] == y[t]^2 - x[t] - 1,
           y'[t] == -x[t ] y[t] - y[t],
          x[0] == #[[1]],
           y[0] == #[[2]]},
          {x[t], y[t]}, {t, 0, 100},
           StartingStepSize -> 0.001,
          MaxStepSize -> 0.005,
          MaxSteps -> ∞]&,
           N[icpts]]

Out[59]= {{{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}},
          {{x[t] → InterpolatingFunction[{{0., 100.}}, <>][t],
            y[t] → InterpolatingFunction[{{0., 100.}}, <>][t]}}}
```

```
In[60]:= ParametricPlot[
          Evaluate[{x[t], y[t]} /. Flatten[solns, 1]], {t, 0, 50},
          PlotRange -> All,
         Axes -> False, Frame -> True]
```




```
Out[60]= - Graphics -
```

```
In[61]:= curves = ParametricPlot3D[
          Evaluate[{x[t], y[t], L[{x[t], y[t]}]} /.
             Flatten[solns, 1]],
          {t, 0, 50},
          PlotRange -> All]
```




```
Out[61]= - Graphics3D -
```

```
In[62]:= Show[curves, lya,
         ViewPoint -> {0.754, -1.392, 2.990}]
```




```
Out[62]= - Graphics3D -
```

This is an attempt at drawing the integral curves of v on the graph of the function L. You can see it move towards the minimum.

■ 3. Apply the Picard Iteration to solve $\frac{dx}{dt} = 1 + x^2$ with $x(0) = 0$. Show that after three Picard iterations, the result agrees with the true solutions for terms of degree five or less in t. Can the solution to this equation be continued for all time?

For Picard iteration, we have the equation $\frac{dx}{dt} = f(x)$, where

```
In[63]:= f[x_] := 1 + x^2
```

The algorithm is to iterate the map

$$\text{In[64]:= Pic[u\_, x\_, t\_] := } x + \int_0^t f[u[\tau]]\, d\tau$$

We can let u_0 be the constant function zero.

```
In[65]:= Pic[0&, 0, t]

Out[65]= t
```

Then apparently u_1 is the indentity function.

```
In[66]:= Pic[Identity, 0, t]
```

Out[66]= $t + \frac{t^3}{3}$

We see a power series in *t* developing.

```
In[67]:= Pic[Function[t, t + t^3/3], 0, t]
```

Out[67]= $t + \frac{t^3}{3} + \frac{2\,t^5}{15} + \frac{t^7}{63}$

Now let's compare this to the analytic answer. *Mathematica* knows how to find the solution.

```
In[68]:= DSolve[x'[t] == 1 + x[t]^2, x, t]

Out[68]= {{x → (Tan[#1 + C[1]]&)}}

In[69]:= (x /. First[%])[t]

Out[69]= Tan[t + C[1]]
```

Or, we can just use separations of variables.

$$\frac{dx}{dt} = 1 + x^2$$

$$\frac{dx}{1 + x^2} = dt$$

Integrating both sides,

$$\int \frac{dx}{1 + x^2} = \int dt$$

$$\tan^{-1}(x) = t + c$$

$$x = \tan(t + c).$$

For us, the constant $c = 0$ will do. In any case,

```
In[70]:= Series[Tan[t], {t, 0, 5}]
```

Out[70]= $t + \frac{t^3}{3} + \frac{2\,t^5}{15} + O[t]^6$

So we see that the two expressions agree up to order five.

Note that as *t* approaches $\frac{\pi}{2}$, $x(t)$ "blows up." So we cannot use this to find a solution for all time.

■ Show that $\frac{dx}{dt} = 3x^{2/3}$ and $x(0) = 0$ has infinitely many solutions given by $x_c(t) = \begin{cases} 0 & t < c \\ (t-c)^3 & t \geq c \end{cases}$ where $c > 0$ is any constant. Does $f(x) = x^{2/3}$ satisfy a Lipschitz condition?

This problem is completely theoretical, but I don't feel like TeX−ing up a separate sheet. That $\dot{x} = f(x)$ is satisfied by all solutions x_c above is clear except at the point c. But that point is okay since

$$\lim_{x \to c^+} \dot{x}(t) = \lim_{t \to c^+} 3(t-c)^2 = 0 = \lim_{t \to c^-} \dot{x}(t)$$

```
In[71]:= Plot[(x^2)^(1/3), {x, -1, 1}]
```




```
Out[71]= - Graphics -
```

Now f had better not satisfy a Lipschitz condition because then we would have a unique solution. Suppose it did. That is, suppose there exists M such that $|f(x) - f(y)| \leq M|x-y|$. Notice that f is increasing on $[0, \infty)$ and differentiable on $(0, \infty)$, and in this region $f'(x) = \frac{2}{3} x^{-1/3}$. For any x and y with $0 < x < y$, we have

$$M \geq \frac{f(x) - f(y)}{x - y} = f'(z)$$

for some z between them. But if we let x and y tend to zero, then so must z, and

$$\lim_{z \to 0^+} f'(z) = \infty.$$

This contradicts the existence of M.

■ 5. Exponentiate and sketch phase portraits for $\frac{d}{dt}\vec{x} = A\vec{x}$. Which are hyperbolic?

All of these will turn out to be hyperbolic. The one you make up may not be.

■ (a) A= $\begin{pmatrix} -1 & -3 \\ 0 & 2 \end{pmatrix}$.

```
In[72]:= A = {{-1, -3}, {0, 2}}

Out[72]= {{-1, -3}, {0, 2}}
```

Here is a nice *Mathematica* linear algebra function.

```
In[73]:= {S, J} = JordanDecomposition[A]

Out[73]= {{{1, -1}, {0, 1}}, {{-1, 0}, {0, 2}}}

In[74]:= MatrixForm /@ %
```

Out[74]= $\left\{\begin{pmatrix} 1 & -1 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} -1 & 0 \\ 0 & 2 \end{pmatrix}\right\}$

```
In[75]:= S . J . Inverse[S]

Out[75]= {{-1, -3}, {0, 2}}
```

So our matrix *A* is diagonalizable with one positive and one negative eigenvalue. Hence there ought to be a saddle point.

To compute e^{At}, we can either use the Jordan decomposition (since we know how to exponentiate a diagonal matrix):

```
In[76]:= MatrixExp[J t]
```

Out[76]= $\{\{E^{-t}, 0\}, \{0, E^{2t}\}\}$

```
In[77]:= S . % . Inverse[S]
```

Out[77]= $\{\{E^{-t}, E^{-t} - E^{2t}\}, \{0, E^{2t}\}\}$

```
In[78]:= % // MatrixForm
```

Out[78]//MatrixForm=

$$\begin{pmatrix} E^{-t} & E^{-t} - E^{2t} \\ 0 & E^{2t} \end{pmatrix}$$

Or we can do it immediately.

```
In[79]:= MatrixExp[A t]
```

Out[79]= $\{\{E^{-t}, -E^{-t}(-1 + E^{3t})\}, \{0, E^{2t}\}\}$

```
In[80]:= Table[MatrixExp[A t] . {Cos[θ], Sin[θ]},
           {θ, 0, 2 π, π / 16}];
```

```
In[81]:= ParametricPlot[Evaluate[
          Table[MatrixExp[A t] . {Cos[θ], Sin[θ]},
          {θ, 0, 2 π, π / 16}]],
          {t, -3, 3},
         Frame -> True, Axes -> False,
         PlotRange -> {{-2, 2}, {-2, 2}},
         AspectRatio -> 1]
```




```
Out[81]= - Graphics -
```

So it is a saddle node.

■ (b) A= $\begin{pmatrix} 2 & 3 \\ 0 & -1 \end{pmatrix}$

In[82]:= **A** = $\begin{pmatrix} 2 & 3 \\ 0 & -1 \end{pmatrix}$

```
Out[82]= {{2, 3}, {0, -1}}
```

Again, we predict a saddle point.

In[83]:=
```
ParametricPlot[Evaluate[
  Table[MatrixExp[A t] . {Cos[θ], Sin[θ]},
  {θ, 0, 2 π, π / 16}]],
  {t, -3, 3},
 Frame -> True, Axes -> False,
 PlotRange -> {{-2, 2}, {-2, 2}},
 AspectRatio -> 1]
```



Out[83]= - Graphics -

■ (c) $A = \begin{pmatrix} -1 & -1 \\ 1 & -1 \end{pmatrix}$

Notice that this matrix represent the complex number $-1 - i$, so we should see an attracting focus.

In[84]:= $A = \begin{pmatrix} -1 & -1 \\ 1 & -1 \end{pmatrix}$

Out[84]= {{-1, -1}, {1, -1}}

```
In[85]:= ParametricPlot[Evaluate[
          Table[MatrixExp[A t] . {Cos[θ], Sin[θ]},
          {θ, 0, 2 π, π / 16}]],
          {t, -3, 3},
         Frame -> True, Axes -> False,
         PlotRange -> {{-2, 2}, {-2, 2}},
         AspectRatio -> 1]
```




```
Out[85]= - Graphics -
```

▪ (d) $A = \begin{pmatrix} 2 & 3 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{pmatrix}$

```
In[86]:= A = (2   3   0
              0  -1   0
              0   0  -1)

Out[86]= {{2, 3, 0}, {0, -1, 0}, {0, 0, -1}}

In[87]:= {S, J} = JordanDecomposition[A]

Out[87]= {{{0, -1, 1}, {0, 1, 0}, {1, 0, 0}}, {{-1, 0, 0}, {0, -1, 0}, {0, 0, 2}}}

In[88]:= MatrixForm /@ %
```

Out[88]= $\left\{\begin{pmatrix} 0 & -1 & 1 \\ 0 & 1 & 0 \\ 1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 2 \end{pmatrix}\right\}$

```
In[89]:= MatrixExp[A t]
```

Out[89]= $\{\{E^{2t}, E^{-t}(-1 + E^{3t}), 0\}, \{0, E^{-t}, 0\}, \{0, 0, E^{-t}\}\}$

```
In[90]:= ParametricPlot3D[Evaluate[
         Flatten[Table[MatrixExp[A t] . {Cos[ϕ] Cos[θ], Cos[ϕ] Sin[θ], Sin[ϕ]},
          {θ, 0, 2 π, π / 4}, {ϕ, 0, 2 π, π / 4}], 1]],
          {t, -5, 5},
          PlotRange -> {{-2, 2}, {-2, 2}, {-2, 2}},
         AspectRatio -> 1]
```




```
Out[90]= - Graphics3D -

In[91]:= {λ, e} = Eigensystem[A]

Out[91]= {{-1, -1, 2}, {{0, 0, 1}, {-1, 1, 0}, {1, 0, 0}}}
```

This is hyperbolic, too. The expanding direction of the system is the eigenvector for the lone positive eigenvalue: $\boldsymbol{e}_3 = \boldsymbol{e}_x$.